

## 8 Работа с системными переменными

### 8.1 Обзор

В этом учебном разделе рассматриваются следующие элементы:

- измерение тактового времени с помощью таймера.

### 8.2 Измерение тактового времени с помощью таймера

Описание  
измерения  
тактового  
времени с  
помощью  
системного  
таймера KUKA



Рис. 8-1

- \$TIMER[1]
- \$TIMER[2]
- ...
- \$TIMER[32]

Системные переменные \$TIMER[№] служат только для измерения временных процессов.



Ввод значений/индикация таймера \$TIMER[№] осуществляется в миллисекундах (мс).

#### Запуск и останов таймера с помощью KRL

- ЗАПУСК: \$TIMER\_STOP[№] = FALSE
- ОСТАНОВ: \$TIMER\_STOP[№] = TRUE



Таймер можно также предварительно устанавливать, запускать и останавливать вручную с помощью окна индикации.

Принцип  
измерения  
тактового  
времени

Предварительная установка таймера

- При поставке таймер установлен на 0 мс
- Таймер сохраняет текущее значение
- Таймер может переставляться вперед или назад на любое значение

```
; Таймер 5 предварительно установлен на 0 мс
$TIMER[5] = 0

; Таймер 12 устанавливается на 1,5 секунды
$TIMER[12] = 1500

; Таймер 4 устанавливается на -8 секунд
$TIMER[4] = -8000
```

#### ■ Сброс и запуск таймера

```
; Сброс таймера 7 на 0 мс
$TIMER[7] = 0
; Запуск таймера 7
$TIMER_STOP[7] = FALSE
```

#### ■ Останов таймера и последующее сравнение

```
; Таймер 7 работает
...
; Остановка таймера 7
$TIMER_STOP[7] = TRUE

; При 10 секундах или больше выполняется ...
IF $TIMER[7] >= 10000 THEN
...
```



Запуск и остановка таймера всегда выполняются с помощью **указателя предварительного выполнения**. Для генерирования останова предварительного выполнения подходит команда

WAIT SEC 0.

Порядок  
действий при  
измерении  
тактового  
времени

1. Выбрать произвольный таймер из 32 возможных элементов времени
2. Предварительно установить/сбросить таймер
3. Запустить таймер с учетом указателя предварительного выполнения
4. Остановить таймер с учетом указателя предварительного выполнения
5. При необходимости промежуточно сохранить текущее тактовое время и заново предварительно установить таймер

```
DEF MY_TIME( )
...
INI
$TIMER[1] = 0 ; сброс таймера 1
PTP HOME Vel=100% DEFAULT

WAIT SEC 0 ; останов предварительного выполнения
$TIMER_STOP[1]=FALSE ; запуск измерения тактового времени

PTP XP1
PTP XP2
LIN XP3
...
PTP X50
PTP HOME Vel=100% DEFAULT

WAIT SEC 0 ; останов предварительного выполнения
$TIMER_STOP[1]=TRUE ; останов измерения тактового времени

; текущее тактовое время промежуточно сохраняется в таймере 12
$TIMER[12] = $TIMER[1]; значение может быть считано в индикации таймера
END
```



### 8.3 Упражнение: измерение тактового времени и оптимизация

#### Цель упражнения

После успешного завершения этого упражнения вы будете в состоянии выполнить следующие операции:

- работа с таймером (инициализация, запуск, останов);
- оптимизация тактового времени.

#### Условия

Для успешного завершения этого упражнения требуется следующее:

- знания о системных переменных \$TIMER[x].

#### Постановка задач

Создать копию программы «Поддон» и присвоить ей имя «Измерение времени». Необходимо измерить три времени. Таймер 1 должен измерить время укладки на поддоны, таймер 2 – процесс разгрузки, таймер 3 – общее время обеих операций. Учитывать, что таймер **не** должен останавливаться или запускаться с предварительным выполнением.

1. Для измерения времени укладки на поддоны использовать \$TIMER[1] и дополнительно сохранить конечное значение в \$TIMER[4].
2. Назвать \$TIMER[1] «Укладка на поддоны - текущее», а \$TIMER[4] «Укладка на поддоны - старое».
3. Для измерения времени снятия с поддонов использовать \$TIMER[2] и дополнительно сохранить конечное значение в \$TIMER[5].
4. Назвать \$TIMER[2] «Снятие с поддонов - текущее», а \$TIMER[5] «Снятие с поддонов - старое».
5. Для измерения всего времени используется \$TIMER[3] и сохраняется в \$TIMER[6].
6. Назвать \$TIMER[3] «Всего текущее», а \$TIMER[5] «Всего СТАРОЕ».
7. Попытаться рационально оптимизировать программу.
8. Проверить программу в режимах T1, T2 и «Автоматика». Следует соблюдать правила техники безопасности в рамках инструктажа.

#### Что следует знать сейчас:

1. Сколько таймеров имеется в системе управления KUKA и как они запускаются?

.....

.....

2. Назовите пример переменной из \$CONFIG.DAT.

.....

.....

